

Paper 226-30

An Introduction to SAS[®] Applications of the Windows Scripting Host

Stephen Hunt, PRA International, Victoria, BC
Tracy Sherman, PRA International, Victoria, BC
Brian Fairfield-Carter, PRA International, Victoria, BC

ABSTRACT

Analysis Programmers in the Pharmaceutical industry are often confronted with a variety of requirements that are peripheral to the more common tasks of data manipulation and summarization. These include supporting multiple output file types (.doc, .rtf, .ps, .pdf), making operating system calls (for example, to retrieve file attribute information), and carrying out file-handling operations. These tasks are often satisfied through more obscure components of Base SAS (DDE, System functions, 'pipe' filename access, x/DOS commands), and through code-writing macros (custom output 'drivers'), that allow SAS to interact with other applications.

This paper introduces the Windows Scripting Host (WSH) as a simple, powerful, and comprehensive tool for meeting all of the requirements listed above. The WSH allows VBScript and Jscript programs to run on the host operating system as stand-alone applications, and thus gives SAS programmers not only an environment in which to program in COM (the 'Component Object Model'), but also to develop dynamic applications. This paper focuses on the fundamentals and practical application of VBScript for running Operating System and File System commands and for instantiating and controlling Windows applications, and on methods for integrating WSH applications with a SAS session. The final section introduces the ActiveX Data Object for writing data between SAS, Excel, and Access.

INTRODUCTION

THE WINDOWS SCRIPTING HOST

The Windows Scripting Host is a component of Windows Operating Systems, introduced with Windows 95, which allows 'Visual Basic Scripting Edition' (VBScript), Jscript, and Perl to be run 'natively on the host operating system'. This paper limits its focus to VBScript; however, the techniques for writing and executing Jscript and Perl would follow those described here. Note that in order to run Perl on the WSH, it is first necessary to install a Perl language interpreter, such as ActiveState ActivePerl or MKS Pscript.

COMMON FILE-TYPE CONVERSION METHODS

File-type conversion methods commonly seen in the SAS world can be roughly divided into 4 categories:

- (1) Those that use data _null_ 'drivers' to write .ps, .rtf or .html code,
- (2) Those that use SAS ODS output destinations,
- (3) Those that use the DDE (Dynamic Data Exchange) interface to interact with Windows applications, and
- (4) Those that run Visual Basic for Applications (VBA) macros in Windows applications.

Each of these methods has its own advantages and disadvantages. WSH applications are an attractive alternative since they can be tested externally to SAS, they take advantage of output drivers and database engines built into target applications, and they provide a very simple interface between SAS and ActiveX automation/COM, allowing for rapid applications development. Scripts written for the WSH can also often be re-used in dynamic HTML applications with little or no modification.

OPERATING AND FILE SYSTEM CALLS

Operating and file system calls in SAS tend to be made through 'Call System', 'x/DOS', 'pipe' filename access, and system function ('Sysfunc') commands. Again, the WSH provides a simple and comprehensive alternative for meeting file handling and 'inter-process communication' requirements that would otherwise be met via these methods.

VISUAL BASIC SCRIPTING EDITION (VBSCRIPT)

VBScript is closely related to VBA, which serves as the standard scripting component to all MS applications. Practically speaking, the major difference between the two is that variables in VBScript aren't 'strictly typed'. Whereas variables in VBA are type-cast (as 'floating point integers', 'boolean', etc.), in VBScript all variables are 'variants', such that values are resolved in a context-dependant way. In other words, the digit '2' may be treated as text or as a number depending on the context.

INTERACTION BETWEEN SAS AND THE WINDOWS SCRIPTING HOST

USING DATA _NULL_ TO WRITE SCRIPTS

SAS can be involved in a WSH automation application in a variety of ways. The first and most obvious way is simply where SAS writes VBScript or Jscript source code. In this context, macro variables are used to substitute specific values each time the macro is called, and the resultant script is static. A common example would be where a path and/or file name was hard-coded into a script via a macro variable.

For example:

```
%macro vbs_(InFile);
  data _null_; file "c:\wutemp\test.vbs";
  put 'dim objWord'; put 'set objWord=CreateObject("Word.Application)';
  str='objWord.documents.Open '||'"'||"&Infile"||'"; put str; run;
%mend vbs_;
%vbs_(c:\test.doc);
```

PASSING ARGUMENTS TO 'PERMANENT' SCRIPTS

The second form of interaction is where SAS runs VBScript or JScript code. Scripts generated through a data _null_ typically do not require arguments, since these values are coded directly into the script; however, stand-alone scripts are usually designed such that variables in the script are set dynamically through arguments passed in the script call.

To run the script shown above, the call from SAS would simply be:

```
x 'c:\wutemp\test.vbs';
```

If the script had been written as a stand-alone, receiving "InFile" as an argument:

```
Dim InString, objWord
  InFile=WScript.Arguments(0)
  Set objWord=CreateObject("Word.Application")
  objWord.documents.Open InFile
```

, then the script call from SAS would look like:

```
x 'c:\wutemp\test.vbs "c:\test.doc"';
```

In other words, arguments are passed to a stand-alone script simply by listing them after the path/file of the script.

USING SCRIPTS TO WRITE AND EDIT SAS PROGRAMS

The third form is in the use of VBScript text-streaming operations to write and edit SAS programs (which can then be executed in a SAS session launched as an ActiveX automation object). For example, this script generates and runs a simple PROC PRINT:

```
Const SasFolder = "C:\SAS\SASUSER"
Dim FSO, SasPth, TextStream
Set FSO = CreateObject("Scripting.FileSystemObject")
Set SasPth = FSO.GetFolder(SasFolder)
Set TextStream = SasPth.CreateTextFile("test.sas")
TextStream.Write("data test;a=1;output;proc print;run;")
TextStream.Close
Dim objSAS
Set objSAS = Wscript.CreateObject("SAS.Application.8")
objSAS.Visible = True
objSAS.Submit("%INCLUDE 'C:\SAS\SASUSER\test.sas'");
```

SWAPPING DATA BETWEEN SAS AND THE SCRIPTING ENVIRONMENT

SAS and the WSH need a way of swapping data back and forth. This would be necessary if, for instance, you wanted a script to respond to SAS output, to read and edit ODS HTML or rtf output, to transfer data between SAS and other applications (Access, Excel), or to provide a list of files and file attributes to SAS. The simplest and most obvious way of doing this is through simple text files: data _null_ and infile/input statements in SAS, and 'text streaming' operations using the File System Object in the scripting environment. Obvious drawbacks to this method are that you can't specify text and numeric data types in the text file, and your reading/parsing program in either environment must be very specific to the structure of the incoming file. A much more powerful method is to make use of the 'ActiveX Data Object' (ADODB Recordset) which enables you to transfer data directly from the native format of one application to the native format of another (well, not quite – Excel does not support ADO, but you can still read data from an Excel file and write it to a SAS Recordset using ADO). Use of the ActiveX data object is illustrated in the last section of the paper.

DYNAMIC HTML

Although dynamic HTML does not strictly speaking make use of the Windows Scripting Host (although it is possible in dHTML scripts to pass commands such as 'PopUp' and CurrentDirectory to a WSH 'Wscript.Shell' object), it is worth mentioning here simply because VBScript and Jscript programs written to function as stand-alone applications on the Windows Scripting Host can usually be dropped verbatim into dHTML applications, and vice versa. For an introduction to dHTML/SAS applications, refer to Fairfield-Carter *et al.* (2005).

SAMPLE APPLICATIONS

WORD-PROCESSOR-READY DOCUMENTS AND FILE TYPE CONVERSION

Outside of data manipulation and summarization, creating word-processor-ready documents and file type conversion are probably the most common programming challenges for SAS programmers. The example below uses a data

null to generate VBScript code, which in turn opens a target text file in Word, formats it, and saves it as a Word document. For simplicity, this macro assumes that SAS output (for example, created via PROC PRINTTO) has been given a '.doc' file extension, though until it has been saved as a Word document it is actually just a text file.


A point of particular interest is in the use of the 'ReloadAs' method. SAS form characters will often cause Word to assume that the file it is loading is 'encoded', and Word will not automatically default to Windows default encoding, but may select something extremely obscure. As a result, fonts and other document features will be incorrect in the output document. For some reason, you can't specify a default encoding when the file is being opened, but you can in the 'ReloadAs' method. In other words, the script must first open the file (incorrectly) and then immediately re-load it with the correct encoding. To further confuse things, the 'ReloadAs' method can't be called for files that have been opened using the 'Add' method (used to open a blank Word document, or open a file as a Word document), but can only be called for files that have been opened using the 'Open' method.

```
%macro vbs_word(outfile);
data _null_;
  attrib str format=$180.;
  file 'c:\sas\sasuser\sas2word.vbs'; /*(text file for vbscript code);
  put 'dim objwd';
  put 'set objwd = wscript.createobject("word.application");/*Create Word instance;
  str='objwd.documents.open("||"&outfile"||")'; put str; /*Open target file;
  put 'ObjWD.ActiveDocument.ReloadAs msoEncodingWestern'; /*Reload;
  put 'objwd.selection.wholestory'; /*Select the entire document;
  put 'with objwd.selection.font'; /*Set font attributes;
  put '.name = "courier new"'; put '.size = 9'; put '.bold = 0'; put 'end with';
  put 'with objwd.activedocument.pagesetup'; /*Set page orientation and margins;
  put '.orientation = 1'; put '.topmargin = 85'; put '.bottommargin = 73';
  put '.rightmargin = 57'; put '.leftmargin = 58'; put 'end with';
  str='objwd.activedocument.saveas("||"&outfile"||")'; /*Save as a word document;
  put str;
  put 'wscript.sleep 2000'; /*stall process for 2 seconds before quitting word;
  put 'objwd.documents.close'; put 'objwd.application.quit';
run;
options noxwait xsync; /*Set synchronous execution of 'x' process;
x "c:\sas\sasuser\sas2word.vbs"; /*(run the script);
%mend vbs_word;
```

A macro call to convert a SAS listing (with '.doc' file extension) to a Word document would look like:

```
%vbs_word(c:\wutemp\test1.doc);
```

In the 'c:\sas\sasuser' directory, you would see the generated script file, where the '.vbs' extension associates the file with the Windows Scripting Host and displays it with the following icon:



For comparison, if instead of generating the script in a data _null_, the script had been written as a 'permanent' stand-alone, values would be assigned in the following way:

```
Dim objWD, InFile
InFile=WScript.Arguments(0)
Set objWD = WScript.CreateObject("Word.Application")
objWD.Documents.Open(InFile)
(etc.)
```

The script call from SAS would then look like this:

```
x 'c:\sas\sasuser\sas2word.vbs "C:\WUTemp\test1.doc";
```

A macro like that shown above can be easily re-tooled to import SAS/GRAPH output into Word. The key instructions for importing SAS/GRAPH cgm output would be:

```
put 'objwd.selection.inlineshapes.addpicture("c:\wutemp\temp.cgm").select'; /*Insert;
put 'objwd.selection.inlineshapes(1).height = 360'; /*Set height;
put 'objwd.selection.inlineshapes(1).width = 580'; /*Set width;
```

SAVING UNDER OTHER FILE TYPES

By supplying a parameter after the 'save as' file reference, you can use Word's output drivers to save under any of the alternative file types in Word's "save as" drop-down list:

- objwd.activedocument.saveas("test.doc"), 1 → saves as MS Word (file type parameter is optional)
- objwd.activedocument.saveas("test.rtf"), 6 → saves as RTF (position 6 in the 'save as' file type list)
- objwd.activedocument.saveas("test.htm"), 8 → saves as HTML

Writing post-script and pdf files is a bit more involved, since these are produced by printing to other output drivers; this process is described in the next section.

THE 'CURRENTDIRECTORY' PROPERTY AND MULTIPLE-FILE PROCESSING; HARD-COPY, POST-SCRIPT AND PDF PRINTING

The example above showed the use of a dynamically-generated script to carry out processing on a single file. A bit of additional code would allow you to pass output file references to this SAS macro via a macro variable (for example, using a system function to return the full path and file name of a 'filename' reference) rather than as hard-coded literal text. An alternative would be to get a reference to the target directory containing all SAS listing files (and/or SAS/GRAPH output files), and use a loop to process each file in that directory having the correct file extension.

The key ingredients to this alternative are the 'File System Object', which is the universal scripting object for file- and operating-system calls, the CurrentDirectory property of the WScript.Shell object (which returns the path name for the directory the script file is sitting in), and string manipulation for working with path/file names. The string manipulation functions used here are:

- StrReverse, which reverses the input string
- Split, which splits an input string at a specified character, and writes the 2 parts to an array
- Lbound, which retrieves the lower bound of an array (the number "1", in this example)
- Trim, which removes leading and trailing blanks
- Replace, which performs text replacement
- "&", the concatenation operator

Although the sample application given here deals with file printing (hard-copy, post-script, and pdf), it could easily be applied to the file type conversion discussed in the previous section, as well as to the task of concatenating multiple output files into a single document. Syntax for the 'PrintOut' method is a bit cryptic; for an explanation, refer to the section below entitled 'Finding Methods and Method Syntax'.

```
Dim objwd, wshshe, fs
Set WshShe = WScript.CreateObject("WScript.Shell")
Set fs= CreateObject("Scripting.FileSystemObject")
Set f = fs.GetFolder(WshShe.CurrentDirectory)
' (The CurrentDirectory property of the WScript.Shell object is passed to the
' GetFolder method of the File System Object, so 'f' is now a reference to the
' folder the script is sitting in)
Set fc= f.files 'The collection of files in the current directory
Set objwd = WScript.CreateObject("Word.Application")
objwd.visible=false ' (This can be run as an invisible process)
For Each fl in fc 'Loop through each file in the collection
  extens=(split(strreverse(fl.name), ".", -1, 1))
  ' (Split the file name into file and extension strings)
  prefix=strreverse(extens(lbound(extens))) 'Get the file extension
  torep=trim(".") & prefix 'Concatenate '.' to file extension
  suffix=trim(replace(fl.name, torep, " ")) 'Just the file name, no extension

  if ucase(prefix)="RTF" then 'Only if file is an rtf file
    size=int(((fl.size)/102.400)*3) 'Ad hoc calibration to estimate a sleep time
    if size<1000 then ' to allow the file to finish printing before
      size=1000 ' closing and going on to the next file
    end if
    pname=f & "\" & suffix & ".pdf" 'Create file name to copy the temporary file
    fname=f & "\" & fl.name 'Get the full path/file name of the rtf file to open
    tname="c:\temp.pdf" 'Temporary/default pdf output file name
    gname="c:\temp.pdf.pdf" 'To cope with an Acrobat PDFWriter idiosyncrasy -
    'always seems to create an additional copy with doubled file extension
    objwd.documents.open(fname) 'Open the rtf file
    objwd.activeprinter="Acrobat PDFWriter" 'Set active printer to pdf device
    objwd.ActiveDocument.PrintOut 0,0,0,tname,,,,,1 'Print to active printer
    wscript.sleep size 'Stall script execution while file prints
    objwd.ActiveDocument.Close() 'Close the rtf file
    wscript.sleep size
    fs.movefile gname, pname 'Move/rename temporary pdf file to final file
    set badfile=fs.getfile(tname) 'Set a reference to the duplicate temporary pdf
    badfile.delete 'Delete the duplicate
  end if
next
```

```
objwd.application.quit()           'Terminate the instance of Word
```

In order to create a post-script file, all you would need to do would be to change `objwd.activeprinter` to point to a post-script driver. For example, if you had installed the .ps device downloaded from the PharmaSUG web site, and called it Acrobat Distiller (Copy 1), then you would set the active printer to "Acrobat Distiller (copy 1)":

```
objwd.activeprinter="Acrobat Distiller (copy 1)" 'Set active printer to .ps device
```

You could then print to this device (you'd probably want to set the file extension of the output file to '.ps' to avoid confusion, though this is optional), and then, if you wanted, run the post-script file in an application such as Adobe PDFWriter, to produce a pdf file:

```
Dim objwd
set objwd=wscript.createObject("Word.Application")
objwd.Documents.open "c:\wutemp\test.doc"
objwd.activeprinter="Acrobat Distiller (Copy 1)"
objwd.ActiveDocument.PrintOut 0,0,0,"c:\wutemp\test.ps"

Dim wshell
Set wshell = WScript.CreateObject("WScript.Shell")
wshell.run Chr(34) & "C:\Program Files\Adobe\Acrobat 4.0\Distillr\AcroDist.exe " &
Chr(34) & "c:\wutemp\test.ps"
```

For more information on command-line syntax and "wshell.run", refer to the section below entitled "Finding Application Object references and Command-line Syntax on the System Registry".

Finally, you could print the file to a hard-copy printer, simply by setting that as the active printer:

```
objwd.activeprinter=\\My Network\My Printer
```

, or, alternately, setting that as the default printer:

```
Dim net
Set net = CreateObject("WScript.Network")
net.SetDefaultPrinter \\My Network\My Printer
```

, and then printing:

```
objwd.ActiveDocument.PrintOut ' (No parameters are necessary)
```

What makes this attractive is that it gives you the opportunity to set a start time for your print job (for example, if you wanted to print at 2:30am). You would specify the start time, the script would calculate the duration between that time and current system time, and the 'Sleep' function would suspend the process (without using up system resources) for that duration:

```
StartHour=InputBox("Enter Hour of Print Start")
StartMin=InputBox("Enter Minute of Print Start")
If StartHour < 12 Then StartHour=StartHour+12 'Assume a.m. hour means next day
If StartHour >= Hour(Time) Then 'If duration is a positive value
  Diff=(StartHour*60*60 + StartMin*60) - (Hour(Time)*60*60 + Minute(Time)*60)
  Diff_=Diff*1000 'Sleep time in milliseconds, as expected by the Sleep function
  Wscript.Sleep Diff_ 'Sleep for the desired duration
Else
  MsgBox("Start time is before current system time - cannot calculate sleep time")
End If
```

As a final point of interest, the CurrentDirectory/file-collection architecture could also be applied to SAS programs:

```
Dim ObjSAS
Set ObjSAS = WScript.CreateObject("SAS.Application.8")
...[get current directory, set up loop, check for .SAS file extension]...
ObjSAS.Submit("%include '" & f & "\" & f1.name & "';")
```

MANIPULATION OF RTF CODE

Rtf code is extremely useful, because it can be parsed and edited as simple text, but rendered as visually complex documents. Occasionally you may run into special characters that need to appear in SAS output that can't be retrieved through the BYTE function. In these cases, it may be convenient to create SAS output as .rtf, with a literal text 'place-holder' (such as '/*Infinity*/'), and then edit the .rtf as a text stream in order to replace the literal with the rtf code to produce the desired symbol (in this case, ∞):

```


Dim FSO, File, textin, textout, string, stringout
Set FSO = CreateObject("Scripting.FileSystemObject")
Set File = FSO.GetFile("c:\wutemp\test.rtf")
Set textin = File.OpenAsTextStream 'Open the source file as a text stream
Set textout = FSO.CreateTextFile("c:\wutemp\test_modified.rtf", True)
' (Open the output file as a text file for writing)
Do While Not textin.AtEndOfStream 'Loop through each line of the source file
string = textin.ReadLine & NewLine 'Read the line
stringout = Replace(string, "/*Infinity*/", "\u8734\'38")
' (Replace literal text with rtf code for desired character)
textout.Write(stringout) 'Write the modified line to the output file
textout.WriteLine("") 'Carriage-return to the next line of the output file
Loop
textin.close 'Close the input text stream
textout.close 'Close the output text stream

```

FINDING METHODS AND METHOD SYNTAX

For methods such as 'PrintOut', the number and order of arguments may not be readily apparent. Often the quickest way to determine expectations of a particular method is by recording a VBA macro in Word (or Excel), and then 'translating' to VBScript (remembering of course that values are stored as 'variants', and that recorded macros will inevitably contain many lines of irrelevant code).

Another useful tool is the 'Object Browser' in Word's Visual Basic Editor, which provides a template for the syntax of each method. The screen-shot below shows details of the 'PrintOut' method, under 'Document'. The output file name is specified in argument 4, and 'print to file' is specified as argument 11. Many arguments can be ignored altogether, and will simply resolve to default values. Arguments such as 'PrintToFile' are Boolean (true/false), or 1/0 as VBScript variants, so referring to the object browser is helpful for deciphering cryptic VBScript method calls.




FINDING APPLICATION OBJECT REFERENCES AND COMMAND-LINE SYNTAX ON THE SYSTEM REGISTRY

In some cases, unless you have sample code to borrow from, some research may be required in order to determine the correct reference for an application object. For example, the statement

```
Set objSAS=Wscript.CreateObject("SAS.Application")
```

may launch an instance of SAS version 8 on some machines but not others. The reason for this is shown by viewing the system registry (which you can launch by typing 'regedit' at the command prompt):



If "SAS.Application" appears in the registry, and 'current version' points to "SAS.Application.8", then the CreateObject call above will work. If "SAS.Application" does not appear, then the CreateObject call must reference "SAS.Application.8".

The system registry is also useful for providing command-line syntax for certain operations. For example, if you have created an instance of SAS with a CreateObject call, you might assume that you could submit some sort of 'objSAS.Open' command in order to open a SAS program, as you would for an instance of Word. SAS does not appear to support such a method, leaving you with the following options, using the 'submit' method:

```
objSAS.Submit("dm pgm 'inc c:\wutemp\test.sas';")
```

or

```
dmstr="inc c:\wutemp\test.sas;"
objSAS.Command(dmstr)
```

, both of which perform unpredictably (as in, they may not always open the target file).

An alternative is to execute the "SASOACT.exe" executable from the WScript.Shell interface (the command syntax can be looked up on the system registry):

```
Dim wshell
Set wshell = CreateObject("WScript.Shell")
wshell.Run Chr(34) & "C:\PROGRA~1\SASINS~1\SAS\V8\CORE\SASEXE\SASOACT.EXE" & Chr(34)
& " action=Open datatype=SASFile filename='c:\wutemp\test.sas'
progid=SAS.Application.8"
```

The 'Chr(34)' function calls return a double-quote, which is useful when you need to double up on the double-quotes (as is required in this example, since the path name to the executable file contains blank spaces, and so must be enclosed in double quotes, while the string passed to the 'Run' method must also be in double-quotes). Of course, the instance of SAS that you now have was not spawned as an ActiveX object, so in order to get an object reference to it you need to call the 'GetObject' method:

```
dim objSAS
on error resume next
While IsObject(objSAS)=False
Set objSAS = GetObject(, "SAS.Application.8")
If IsObject(objSAS)=True Then
objSAS.Submit("%let testvar = TESTVALUE;")
End If
Wend
```

As you can see, this is not without its own set of problems, since you need to wait until the application instance exists before you can create a reference to it and start submitting program statements. Interestingly, the fact that this works appears to contradict SAS Institute claims that you can't get an object reference to an instance of SAS that was not spawned as an ActiveX process.

FILE HANDLING

Previous examples have illustrated the use of the File System Object for retrieving file and path names, for looping through collections of files, for reading and writing text files, and for copying, moving, re-naming and deleting files. Additional capabilities include retrieving file attribute information (size, date created, etc.) as properties of the 'file' object, creating directories and subdirectories, and testing for the existence of target files.

ADODB RECORDSET – USES OF THE ACTIVEX DATA OBJECT

INTRODUCTION

Although the use of simple text files is an inviting starting point for swapping data between SAS and the scripting environment, the 'ActiveX Data Object' (ADO) is by far the preferred method, first for the obvious reason that you don't have to manage stray text files, but more importantly because it provides a very powerful data interchange between SAS and other applications such as Excel and Access. With very little work, you can create your own SAS <--> Excel and SAS <--> Access import/export tools, tailored with any additional data manipulation steps you might care to add. A key point is that you can do this without necessarily creating an instance of either application.

Working with ADO is similar to working with other 'application objects' described so far – you create an instance of 'ADODB.RecordSet' by using the 'CreateObject' method, and call methods in this object. However, the requirements for a data interchange also include a database connection ('ADODB.Connection'), also created using 'CreateObject' and qualified by a specific 'connection string'. The connection string indicates the data 'provider' (the 'engine' that reads and writes the particular data files) and the data source.

CREATING SAS DATA

To illustrate, say you wanted to create a SAS dataset. First you would create instances of the necessary objects:

```
Dim RecSet, Connection
Set RecSet = WScript.CreateObject("ADODB.Recordset")
Set Connection = WScript.CreateObject("ADODB.Connection")
```

Then you would use the ADO.Connection object to connect to SAS (specifically, to SAS's database engine) by passing a connection string as an argument to the 'Open' method, and set the 'ActiveConnection' property of the ADO.RecordSet object to this connection:

```
Connection.Open "Provider=SAS.IOMProvider.1;Data Source=_LOCAL_"
RecSet.ActiveConnection = Connection
```

Now all you need to do in order to create the SAS dataset is pass an SQL sentence (as literal text) as an argument to the 'Execute' method on the active connection:

```
Connection.Execute "create table sasuser.recset (test_text char(20), test_num num) "
```

This creates a SAS dataset with 2 variables, the first character with length 20, and the second numeric, and with no records. In order to populate this dataset, we first need to open it for writing on the active connection:

```
RecSet.Open "sasuser.recset", Connection, 2, 4, 512
```

Next we need to create records ('AddNew'), assign values to variables, and add the records to the data set ('UpdateBatch'):

```
Dim i
i=10
Do Until i = 0
  RecSet.AddNew
  RecSet.Fields(0).Value="number" & i
  RecSet.Fields(1).Value=i
  RecSet.UpdateBatch
  i = i - 1
Loop
```

Finally, say we wanted this data set to be re-ordered in increasing values of the numeric variable – all we'd need to do would be to call the 'Execute' method again with the appropriate SQL sentence:

```
Connection.Execute "create table sasuser.recset1 as select * from sasuser.recset
order by test_num"
```

WRITING EXCEL DATA TO SAS

Now that we have the basics of creating and manipulating SAS data, we can move into some practical examples. Probably the first and most obvious one is that of writing Excel data to a SAS dataset. From the code above, we can see that we need to know

1. The location of the source data, and how to read it
2. The variables and attributes of the source data, to plug into the first 'Create Table' statement
3. The number of records on the source data, to determine the size of the loop used to populate the SAS data set.

Because Excel lacks the database engine component of a true database application, data can only be read by first creating an instance of the application (which, fortunately, can be invisible), then opening the target spreadsheet, and reading values via row/column references. For the sake of simplicity, the file reference and cell range have been hard-coded into this example, though you will no doubt recognize the potential for passing these in as arguments. (The potential for determining the cell range 'dynamically', by parsing rows and columns to determine the starting point of missing values on each dimension, should also be apparent.)

First, create the ADO objects and open the connection:

```
Dim RecSet, Connection
Set RecSet = WScript.CreateObject("ADODB.Recordset")
Set Connection = WScript.CreateObject("ADODB.Connection")
Connection.Open "Provider=SAS.IOMProvider.1;Data Source=_LOCAL_"
RecSet.ActiveConnection = Connection
```

Next, create an instance of Excel, and select the target data (column A row 1 to column F row 10) in the target workbook page:

```
Dim appXL
Set appXL = WScript.CreateObject("Excel.Application")
Set GetBook = appXL.Workbooks.Open("C:\WUTemp\xls2sas\test.xls")
Dim GetSheet, GetRange
On Error Resume Next
Set GetSheet = GetBook.Sheets("Sheet1")
Set GetRange = GetSheet.Range("A1:F10")
```

Next, starting at row 2, parse each column to determine an appropriate data type (it is assumed that row 1 contains the variable names, with no blank spaces such that they can be used as SAS variable names):

```
Function ParseColumn(Col)
  Dim Lngth, Lngth_test, Typ
  Lngth=8 'Start with a default of numeric, length 8
  Typ="num"
  For Rw = 2 to GetRange.Rows.Count - 1 'For every value in the column
    Test = GetRange.Cells(Rw,Col).Value 'Capture the value
    Lngth_test = Len(Test) 'Capture the length of the value
    If Lngth_test > Lngth Then Lngth = Lngth_test 'Capture the greatest length
    If Typ="" or Typ="num" Then
```



```

    If IsNumeric(Test)=False Then
        Typ="char"      'If ANY non-numeric found, set type to character
    End If
End If
Next
If Typ="num" Then Lngth=8
ParseColumn=Typ & ":" & Lngth 'Return a 'type:length' string
End Function
'Call the function for each column
Dim Typ, Lngth '(note that arrays must be 'ReDim'ed to the desired size)
ReDim Typ(GetRange.Columns.Count-1) 'Array to hold appropriate data type for each var
ReDim Lngth(GetRange.Columns.Count-1) 'Array to hold appropriate length for each var
For i = 1 to GetRange.Columns.Count
    RString=ParseColumn(i) 'Get the 'type:length' return value
    Typ(i)=Left(RString, Instr(ParseColumn(i),":")-1) 'Pull off the type...
    Lngth(i) = Right(RString, Len(RString)-Instr(ParseColumn(i),":")) '...and length
Next

```

Using variable names captured from row 1 and attributes determined by parsing each column, generate and execute an SQL sentence to create a shell SAS data set:

```

sql = "create table sasuser.recset ("
bFirst=True
For j = 1 to GetRange.Columns.Count
    If bFirst=False and j<GetRange.Columns.Count Then sql = sql & ", "
    bFirst = False
    sql = sql & " " & GetRange.Cells(1,j).Value & " " & Typ(j) & "(" & Lngth(j) & ")"
Next
sql = sql & ")"
Connection.Execute sql

```

Finally, open and populate the SAS data set:

```

RecSet.Open "sasuser.recset", Connection, 2, 4, 512
For i = 2 To GetRange.Rows.Count 'For each row
RecSet.AddNew 'Add a record
    For j = 1 To GetRange.Columns.Count 'For each column
        RecSet.Fields(j - 1) = GetRange.Cells(i,j).Value 'Assign the value
    Next
RecSet.UpdateBatch 'Update the SAS data set with the newly populated record
Next
appXL.application.quit 'Terminate the instance of Excel

```

WRITING ACCESS DATA TO SAS

The process of writing Access data to a SAS dataset is quite similar, but streamlined in that you don't need to launch an instance of Access, and variable names and attributes are provided as properties of RecordSet.Field. The major steps are:

1. Create RecordSet and Connection objects for SAS (as above)
2. Create RecordSet and Connection objects for Access, the only difference being the connection string, and the absence of a data library reference required with SAS:


```

Connection.Open "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=C:\WUTemp\test.mdb"
AccessRecordSet.Open "test", Connection, 2, 4, 512

```
3. Capture variable names and attributes from the Access table, and use these to create the shell SAS data set (note that numeric codes for different Access data types can be determined by using a MsgBox(field.Type) function call for each field in RecordSet.Fields):

```

bFirst = True
sql = "CREATE TABLE " & SASName & " ("
For Each fld In AccessRecordSet.Fields
    If Not bFirst Then sql = sql & ", "
    sql = sql & fld.name
    If fld.Type=129 Or fld.Type=200 Or fld.Type=130 Or fld.Type=202 Then
        sql = sql & " CHAR(" & fld.DefinedSize & ")"
    ElseIf fld.Type=5 Or fld.Type=11 or fld.Type=3 Then
        sql = sql & " NUM"
    ElseIf fld.Type=7 Then
        sql = sql & " DATE"
    Else
        sql = sql & " Char(255)" '(If no attribute is provided, force as text)
    End If

```

```

    bFirst = False
  Next
  sql = sql & " "
4. Read values from the Access table, and write to the SAS data set (note that SAS and Access use different
   default arbitrary cut-off dates when storing date values, so a conversion is required):
  Dim col
  While Not AccessRecordSet.EOF
    SASRecordSet.AddNew
    For col = 0 To AccessRecordSet.Fields.Count - 1
      If AccessRecordSet.Fields(col).Type=7 Then 'If source value is a date...
        SASRecordSet.Fields(col).Value=AccessRecordSet.Fields(col).Value - 21916
        '(Convert to a SAS date equivalent (different cut-off dates!!))
      Else 'Otherwise, just use the value as is
        SASRecordSet.Fields(col).Value=AccessRecordSet.Fields(col).Value
      End If
    Next
    SASRecordSet.UpdateBatch
    AccessRecordSet.MoveNext
  Wend

```

FURTHER APPLICATIONS OF ADO

Although the above examples show the process of writing to SAS datasets, in a way that mimics PROC IMPORT, the direction can easily be reversed to mimic PROC EXPORT. One thing to note is that SQL syntax in Access has its own peculiarities, so some adjustment is necessary if you are used to working with PROC SQL in SAS. ADO can also be used for moving data between SAS and other native formats, notably Oracle. Data manipulation in ADO offers quite a number of interesting possibilities: text files can be written to record sets, and SQL then used for processing (for example, to edit rtf code, and to do things like parse SAS log files for errors and warnings). A particularly interesting instance of rtf manipulation is in modifying ODS/rtf output to switch floating headers and footers to body text. In addition, combining multiple Access tables or Excel spreadsheets can be done in a single export step, rather than importing each table to SAS and then merging.

CONCLUSION

The Windows Scripting Host provides a powerful and comprehensive tool for all of the 'secondary' problems encountered by a SAS programmer, including file handling (moving, copying, renaming, deleting), creating word-processor-ready output, and supporting alternate file types (post-script, pdf, html, rtf, etc.). ActiveX Data Object programming extends this list to include transferring data between SAS and other native formats such as Excel and Access.

REFERENCES

Fairfield-Carter, Brian, Sherman, Tracy, and Hunt, Stephen (2005), "Instant SAS® Applications with VBScript, Jscript, and dHTML", Proceedings of the Thirtieth Annual SAS Users Group International Conference.

ACKNOWLEDGMENTS

The authors wish to thank first and foremost Jeff Carter of Equinox Software Design (www.equinox.ca) for introducing the possibilities offered by the Windows Scripting Host and for providing many helpful insights and programming tips. We would also like to thank Cara, Kelly, Gavin, Kurtus, PJ, Nicholas, our friends in Biostatistical Services, and Tim Williams for their consistent encouragement and support.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Stephen Hunt, Tracy Sherman, Brian Fairfield-Carter
 PRA International
 600-730 View Street
 Victoria, BC (CAN) V8W 3Y7
 Email: HuntStephen@PRAIntl.com, ShermanTracy@PRAIntl.com, FairfieldCarterBrian@PRAIntl.com
 Web: www.prainternational.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.